Modeling Cultural Assimilation Using Global Migration, Economic Integration, and Cultural Diversity Metrics

Connor Cruz*

NCSSM Online
North Carolina School of Science and Mathematics
Durham, North Carolina

December 10, 2025

Revised August 14, 2026

Abstract

Cultural assimilation is often assumed to be shaped by contemporary forces such as increasing globalization, international migration, and economic integration. This study investigates the extent to which demographic and economic variables can explain variation in cultural diversity. Using Mathematica's `CountryData` resource, we constructed a dataset of linguistic diversity, religious diversity, population, geographic size, GDP per capita, trade value, and net migration rates, yielding 210 countries with complete data. We applied logarithmic transformations to reduce skewed distributions and built a religious fractionalization index following Alesina et al. [1]. Multiple linear regression models were then used to assess the influence of migration and economic integration on cultural diversity while controlling for population and area, with variance inflation factors used to diagnose collinearity among the size-related predictors. Geographic area emerged as by far the strongest predictor of linguistic diversity, with an estimated elasticity of 0.33 and $p \approx 4 \times 10^{-38}$. Economic integration showed no robust association with either diversity measure: an apparent effect of GDP per capita disappeared once collinear size predictors were removed. Migration rate was weakly and positively associated with linguistic diversity, the opposite of what an assimilation account would predict. Religious fractionalization was only weakly explained by any of the variables considered, and its relationship with geographic area was negative rather than positive. These results suggest that linguistic diversity is shaped primarily by deep geographic and historical forces rather than by contemporary globalization, and they reinforce the importance of computational methods in testing widely held assumptions about cultural change.

Key words: cultural assimilation, migration, religious fractionalization, linguistic diversity, computational social science, computational science, regression modeling, globalization

Revision note: this version corrects an error in the original December 2025 analysis, in which the response variable was passed in the first rather than the last column of the data matrix, causing log area to be fit as the dependent variable in both models. The results, discussion, and conclusions have been revised accordingly, and collinearity diagnostics and an alternative specification for the language count have been added. The dataset and diversity measures are unchanged.

1 Introduction

Cultural assimilation is a central theme in the study of globalization. As nations become increasingly interconnected through trade, migration, and communication technologies, societies experience new forms of cultural contact that influence languages, religions, values, and social identity. Economic integration fosters shared markets and common communication systems, often diffusing cultural norms across borders [3]. For example, English has become the global language of business due to the economic prominence of the United States, shaping linguistic expectations even in non-English-speaking nations.

Despite this, cultural assimilation through economic forces is not universal. Some countries deliberately resist cultural homogenization. For instance, France has long adopted cultural protection policies to preserve linguistic and artistic traditions [5], demonstrating that globalization pressures can be met with both adoption and resistance.

Migration serves as another powerful mechanism of cultural change. When people cross borders for work, education, or refuge, they bring their languages, religions, and customs with them. Classic examples include the migration of Turkish laborers to Germany under the *Gastarbeiter* program and the growth of bilingual Hispanic communities in the United States. These cases illustrate that migration rarely produces complete assimilation; instead, it generates hybrid identities, a dynamic documented extensively in acculturation research [4].

Although both economic integration and migration appear to influence cultural assimilation, their impact varies widely across countries and remains difficult to quantify. Some of the most culturally diverse nations have minimal modern migration, while highly immigrant-receiving countries can remain relatively homogeneous. This mismatch raises an important empirical question: to what extent do contemporary forces such as migration and economic integration actually explain the cultural diversity observed across countries today?

To investigate this question, we use a computational approach combining global economic, demographic, and cultural data from Mathematica's `CountryData` resource. We analyze two dimensions of cultural diversity: linguistic diversity and a newly constructed religious fractionalization index. We then apply multiple linear regression to test whether migration rate, GDP per capita, and trade value significantly predict either form of diversity after controlling for population and geographic area. We then evaluate how these results align with common assumptions about globalization and assimilation.

2 Computational Approach

The data used in this study were gathered from Mathematica's `CountryData` resource [6]. We obtained the following variables for each country: GDP per capita, trade value, net migration rate, population, geographic area, languages spoken, religions, and religious population shares.

Certain variables exhibited strong positive skew, particularly trade value and population. To mitigate the influence of outliers, four variables were logarithmically transformed: GDP per capita, trade value, population, and geographic area. Since trade value may be zero for some countries, we defined

$$\text{LogTradeValue} = \log(1 + \text{TradeValue}). \tag{1}$$

To measure cultural diversity, and therefore cultural assimilation, two quantities were analyzed: linguistic diversity and religion.

Linguistic diversity was quantified simply as the number of languages listed for each country, an approach that aligns with prior large-scale cultural analyses [2]. Although this is a simple metric, this measure effectively captures significant differences between countries with only one or a few dominant national languages and those with potentially hundreds of indigenous languages.

Religious diversity was measured using the widely used religious fractionalization index (RFI), an adaptation of the Herfindahl concentration index [1]. The RFI represents the probability that two randomly selected citizens adhere to different religions. An RFI close to 0 would correspond to less religious diversity, while a higher RFI would imply more religious diversity among the population.

Equation (2) gives the religious fractionalization index, where n represents the total number of religions in a population and π_i is the share of the population which practices religion i:

$$\mathrm{RFI} = 1 - \sum_{i=1}^{n} \pi_i^2. \tag{2}$$

In addition to constructing diversity measures, several preprocessing steps were also required to ensure that the data were able to be statistically modeled. Many values obtained from `CountryData` are expressed as `Quantity` objects with units attached, which cannot be directly analyzed in numerical computations. To address this issue, all quantities were converted to unit-free numerical values using the `QuantityMagnitude` function. Additionally, missing or incomplete entries were removed to prevent inaccurate data points, allowing relationships to be expressed more factually.

The data were finally combined into association structures, which allowed the regression models to identify each country's data as observations and ensured consistent analysis of a country's provided data. Of the 249 entities returned by `CountryData`, 210 had complete data across all required variables and form the sample for the linguistic diversity model. A further 17 lacked usable religious population shares, leaving 193 countries for the religious fractionalization model.

Two multiple linear regression models were constructed. The first used the number of languages as the dependent variable, while the second used RFI. Both models used the same predictors: migration rate, LogGDPPerCapita, LogTradeValue, LogPopulation, and LogArea.

A multiple linear regression framework was selected for this study because it provides an interpretable method for analyzing the relationship between cultural diversity metrics and their respective predictors. Linear models allow the unique contribution of each variable to be isolated while the others remain constant, thus allowing for multiple correlative factors to be analyzed. Although cultural occurrences are often complex and shaped by nonlinear processes, linear regression still offers a useful approximation of relationships and is widely used in studies on cultural and economic diversity [1, 2]. Because each country represents an independent geopolitical unit, the assumption of independent observations is reasonably justified.

2.1 Model specification

Two specification problems became apparent during analysis, and both are reported here because each materially changes the conclusions.

Skew in the response. The number of languages is a count ranging from 1 to 839 with pronounced right skew. Fitting it directly by ordinary least squares produced fitted values as low as -76.5, which is impossible for a count. A Poisson model was considered as the conventional alternative, but the data are severely overdispersed: the ratio of Pearson residual variance to the mean is 86.5, far above the value of 1 that Poisson regression assumes. Poisson standard errors would therefore be badly understated. We instead model log(number of languages), which accommodates the skew, keeps the response strictly positive, and yields coefficients interpretable as elasticities. As every country reports at least one language, no zero-count adjustment is required. A negative binomial model, which permits variance to exceed the mean, would be the more principled alternative and is left to future work.

Collinearity among size predictors. Log population, log area, and log trade value are all proxies for country size and are strongly intercorrelated: $r = 0.859$ between log population and log area, and $r = 0.788$ between log trade value and log population. Variance inflation factors for the full specification are reported in Table 1. Values above 5 are conventionally treated as problematic and values above 10 as severe; log population and log trade value exceed 25.

Table 1: Variance inflation factors, full specification.

Predictor	VIF
Migration rate	1.15
Log GDP per capita	10.70
Log trade value	25.04
Log population	29.58
Log area	3.98

Under collinearity of this magnitude the individual coefficients are not interpretable, since correlated predictors absorb one another's explanatory power and inflate each other's standard errors. We therefore report a reduced specification retaining migration rate, log GDP per capita, and log area, and dropping log population and log trade value. The reduction was decided on the basis of the diagnostics rather than on the significance of any result, and both specifications are reported below so the effect of the reduction is visible. In the reduced specification all VIFs fall to between 1.13 and 1.24.



Figure 1: Migration rate versus number of languages.



Figure 2: Migration rate versus religious fractionalization.

3 Results and Discussion

3.1 Migration and Cultural Diversity

To visualize the relationship between migration rate and cultural diversity, we created two scatterplots. Figure 1 displays migration rate versus the number of languages spoken in each selected country. As shown by the figure, the points cluster tightly around a migration rate of zero. However, the number of languages ranges from one to over 800. There does not appear to be any linear trend, nor does there seem to be any trend in general.

Countries with high linguistic diversity do not seem to exhibit unusual or significantly high migration rates. Moreover, countries with low linguistic diversity occur across all migration levels.

A similar pattern is evident in Figure 2, which plots migration rate against religious fractionalization. Migration rates share the same pattern as above, being centered around zero, while religious diversity spans nearly the entire range of 0 to 1. Countries of high fractionalization, as well as countries with relatively homogeneous fractionalization, mostly appear with a migration rate near 0. In general, there is not much clustering nor a significant relationship between the data.

These scatterplots ultimately support the conclusion that modern migration rates do not meaningfully correspond to levels of cultural diversity. Any variation in diversity, whether in languages or religion, appears independent of contemporary migration.

3.2 Linguistic Diversity Regression Results

Fitting the raw language count by ordinary least squares against all five predictors yields no significant coefficients at $\alpha = 0.05$ and $R^2 = 0.203$. As noted in Section 2.1, this specification is compromised on two fronts: fitted values extend to

Table 2: Reduced model for log(number of languages), $N = 210$.

	Estimate	Std. Err.	t	p
(Intercept)	-0.6967	0.5099	-1.3664	0.1733
mr	27.7802	11.8148	2.3513	0.0197
lgdp	0.0176	0.0442	0.3988	0.6904
la	0.3309	0.0206	16.042	4.1×10^{-38}
$R^2 = 0.576$				

Table 3: Reduced model for religious fractionalization, $N = 193$.

	Estimate	Std. Err.	t	p
(Intercept)	0.7538	0.1528	4.9330	1.8×10^{-6}
mr	3.0793	3.5752	0.8613	0.3902
lgdp	-0.0167	0.0131	-1.2706	0.2054
la	-0.0172	0.0063	-2.7372	0.0068
$R^2 = 0.043$				

-76.5, and three predictors carry variance inflation factors above 10. Its null results should not be read as evidence of absence.

Modelling log(number of languages) raises the fit substantially, to $R^2 = 0.644$ on the full predictor set. In that specification migration rate ($p = 0.042$), log GDP per capita ($p = 0.034$), and log trade value ($p = 0.010$) all reach significance. However, the apparent GDP effect does not survive the removal of the collinear size predictors: in the reduced specification of Table 2, log GDP per capita falls to $\hat{\beta} = 0.018$ with $p = 0.690$. It was absorbing variance properly attributable to country size rather than measuring an independent effect of wealth.

Log area dominates the reduced model, with $\hat{\beta} = 0.331$ and $p \approx 4.1 \times 10^{-38}$. Because both sides are logged, this coefficient is an elasticity: a one percent increase in land area is associated with roughly a third of a percent increase in the number of languages. Fitted alone against the raw count, log area accounts for $R^2 = 0.162$ of the variance, against 0.203 for the full five-predictor model — that is, nearly all of the model's explanatory power comes from geographic area. This is consistent with patterns exemplified by geographic isolation in the past, where mountain ranges, islands, and sheer distance allowed languages to diverge and persist.

Migration rate remains significant in the reduced model ($\hat{\beta} = 27.78$, $p = 0.020$), and its sign is positive. Countries with higher net migration are associated with more languages, not fewer. The effect is modest — across the observed migration range of roughly -0.02 to 0.023, the fitted difference is on the order of a threefold change in language count — and it is the weakest of the surviving results, moving from $p = 0.215$ in the count model to $p = 0.020$ here. It should be treated as suggestive. However, the direction is notable: an assimilation account would predict a negative association, and the data show the reverse.

Log GDP per capita is not significantly associated with linguistic diversity in any specification that controls collinearity.

3.3 Religious Fractionalization Regression Results

Table 3 presents the reduced model for the religious fractionalization index.

Log area is the only significant predictor ($p = 0.0068$), but its coefficient is negative: each e-fold increase in land area is associated with a decrease of roughly 0.017 in RFI. Fitted alone, log area gives $\hat{\beta} = -0.0153$, $p = 0.011$, $R^2 = 0.034$.

This runs against the intuition that larger countries should encompass more religious groups and therefore exhibit more balanced religious distributions. The data do not support that expectation. One plausible reading is that the two diversity measures are not analogous: languages accumulate through isolation and can persist in small communities indefinitely, whereas religions spread through conversion and institutional networks, so a large territory may be consolidated under a single dominant faith more readily than under a single language. We note this as a possible interpretation rather than a tested mechanism.

Neither migration rate nor GDP per capita is significant. More importantly, the model as a whole explains only 4.3% of the variance in religious fractionalization. Whatever determines a country's religious composition is largely absent from the demographic and economic variables considered here.

4 Conclusions

This study tested whether contemporary migration rates and economic integration predict linguistic and religious diversity across countries. Using a global dataset from Mathematica's `CountryData` resource [6] and diversity metrics grounded in established cultural theory [1, 2], we constructed multiple linear regression models over 210 countries for linguistic diversity and 193 for religious fractionalization.

Three findings emerge. First, geographic area is by far the strongest predictor of linguistic diversity. In a log-log specification it carries an elasticity of 0.33 with $p \approx 4 \times 10^{-38}$, and fitted alone it accounts for most of the explanatory power of the full model. Larger countries hold more languages, consistent with a long history of geographic isolation permitting linguistic divergence.

Second, economic integration shows no robust relationship with cultural diversity. GDP per capita appeared significant in one specification but that result did not survive the removal of collinear size predictors, and trade value was significant only in a specification where its variance inflation factor exceeded 25. Neither is defensible as an independent effect.

Third, migration rate is weakly but positively associated with linguistic diversity. This is the opposite of what an assimilation account predicts, and although the effect is modest and only marginally significant, its direction gives no support to the view that contemporary migration produces cultural homogenization. It is more consistent with the acculturation literature's finding that migration generates hybrid identities rather than replacement [4].

Religious fractionalization behaves differently from linguistic diversity in two respects. Its relationship with geographic area is negative rather than positive, and the full model explains only 4.3% of its variance. The determinants of religious composition appear to lie largely outside the demographic and economic variables considered here.

Taken together, these results suggest that cultural diversity in the present world is not heavily shaped by modern globalization forces such as migration and economic integration. Linguistic diversity in particular reflects long-term geographic and historical processes that persist regardless of recent migratory or economic change, as geographic boundaries most often remain unchanged.

Several limitations bear on these conclusions. Counting languages treats a national language and a small indigenous language as equivalent units, and a diversity index weighted by speaker population would capture linguistic concentration more faithfully, as the RFI does for religion. The models are cross-sectional, so they identify association rather than change over time; a panel design would be required to test assimilation as a process. The remaining predictors are correlated with unmeasured historical factors such as colonial history and state formation, which the present specification cannot separate from area itself, and a negative binomial model would handle the language count more rigorously than the log transformation adopted here.

Future work may incorporate additional predictors such as urbanization, political structure, and internal migration, and could develop a diversity metric that normalizes for geographic area directly. Nevertheless, these findings demonstrate that computational modeling can uncover the structural forces shaping cultural patterns and challenge assumptions about globalization and cultural assimilation.

Beyond these findings, this study highlights a broader insight on analyzing cultural dynamics: that quantitative approaches can reveal when widely assumed cultural relationships lack empirical support. Globalization is often portrayed as an indomitable force producing uniformity, yet these results suggest that contemporary demographic and economic forces exert far less influence on cultural diversity than is commonly believed. The analysis also illustrates how readily specification choices can manufacture or conceal a result: the significance of GDP per capita depended entirely on which collinear predictors remained in the model, and the apparent explanatory power of an earlier specification proved to be an artifact of how the model was framed. Reporting such diagnostics is therefore not incidental to computational social science but central to it.

Acknowledgments

The author thanks Mr. Robert Gotwals for assistance with this work.

References

[1] Alesina, Alberto, Arnaud Devleeschauwer, William Easterly, Sergio Kurlat, and Romain Wacziarg. "Fractionalization." *Journal of Economic Growth*, vol. 8, no. 2, 2003, pp. 155–194.

[2] Fearon, James D. "Ethnic and Cultural Diversity by Country." *Journal of Economic Growth*, vol. 8, no. 2, 2003, pp. 195–222.

[3] Guiso, Luigi, Paola Sapienza, and Luigi Zingales. "Does Culture Affect Economic Outcomes?" *Journal of Economic Perspectives*, vol. 20, no. 2, 2006, pp. 23–48.

[4] Berry, John W. "Immigration, Acculturation, and Adaptation." *Applied Psychology: An International Review*, vol. 46, no. 1, 1997, pp. 5–34.

[5] Hobsbawm, Eric. *Nations and Nationalism Since 1780: Programme, Myth, Reality.* Cambridge University Press, 1990.

[6] Wolfram Research, Inc. *Wolfram Mathematica, Version 14.3*. Champaign, IL: Wolfram Research, Inc., 2024.

Appendix: Analysis Code

The following code was used for data acquisition, cleaning, transformation, visualization, diagnostics, and model fitting. All computations were conducted in Wolfram Mathematica 14.3.

Note the ordering of the columns emitted for each regression: `LinearModelFit` reads the response variable from the last column of the data matrix, so the diversity metric is placed at the end of each row rather than the beginning.

Data Acquisition and Cleaning

```
(* Gets all countries *)
allCountries = CountryData[];

(* Economic and migration information *)
getEconMig[ct_] := <|"Country" -> ct, "Name" -> CountryData[ct, "Name"],
   "GDPPerCapita" -> CountryData[ct, "GDPPerCapita"],
   "TradeValue" -> CountryData[ct, "TradeValueAdded"],
   "MigrationRate" -> CountryData[ct, "MigrationRateFraction"],
   "Population" -> CountryData[ct, "Population"],
   "Area" -> CountryData[ct, "Area"]|>;

(* Cultural information: languages and religions *)
getCultural[ct_] := Module[
   {languages = CountryData[ct, "Languages"],
    religions = CountryData[ct, "Religions"]},
   <|"NumLanguages" -> Length[languages],
     "NumReligions" -> Length[religions]|>];

(* Combine economic/migration and cultural data for each country *)
combinedData =
  Table[Join[getEconMig[ct], getCultural[ct]], {ct, allCountries}];

(* Remove entries with missing core values *)
keys = {"GDPPerCapita", "TradeValue", "MigrationRate", "NumLanguages",
   "NumReligions", "Population", "Area"};
cleanData = Select[combinedData, FreeQ[Lookup[#, keys], _Missing] &];
dataset = Dataset[cleanData];

(* --- Sample size reporting ------------------------------------------ *)
(* The paper needs the post-deletion N, not the raw country count.      *)
Print["Countries returned by CountryData: ", Length[allCountries]];
Print["Countries after listwise deletion: ", Length[cleanData]];
```

Log Transformation

```
(* Function to get magnitude given a string *)
num[x_] := If[QuantityQ[x], QuantityMagnitude[x], x];

(* Add log-transformed variables to each country association *)
modelData = Map[Function[assoc, Module[
     {
      gdp = num[assoc["GDPPerCapita"]], trd = num[assoc["TradeValue"]],
      pop = num[assoc["Population"]],
      area = num[assoc["Area"]],
      mig = assoc["MigrationRate"]
      },
     Association[assoc,
      "LogGDPPerCapita" -> If[gdp > 0, Log[gdp], Missing["NotPositive"]],
      "LogTradeValue" ->
       If[trd >= 0, Log[1 + trd], Missing["NotPositive"]],
      "LogPopulation" -> If[pop > 0, Log[pop], Missing["NotPositive"]],
      "LogArea" -> If[area > 0, Log[area], Missing["NotPositive"]],
```

```
      "MigrationRateNumber" -> num[mig]]
     ]
    ]
   , cleanData];

modelDataset = Dataset[modelData];
```

Religious Fractionalization Index Calculation

```
(* Religious fractionalization index calculation *)
relFractionalization[ct_] := Module[
   {fracs, vals, norm},
   (* Fractions are usually rules: religion -> fraction *)
   fracs = CountryData[ct, "ReligionsFractions"];

   (* Checks if data is not available *)
   If[fracs === Missing["NotAvailable"] || fracs === {} ||
     fracs === Null,
    Return[Missing["NotAvailable"]];
    ];

   (* Extract numeric fractions and clean data *)
   vals = Cases[fracs, (_ -> p_?NumericQ) :> p, Infinity];
   vals = DeleteMissing[vals];

   (* No usable fractions *)
   If[vals === {}, Return[Missing["NotAvailable"]];];

   (* Ensures 0 diversity for 1 religion *)
   If[Length[vals] == 1, Return[0];];

   (* Normalize and compute RFI *)
   norm = vals/Total[vals];
   1 - Total[norm^2]];

(* Add religious fractionalization index to data *)
dataWithRFI =
  Map[Function[assoc,
    Association[assoc,
     "RelFracIndex" -> relFractionalization[assoc["Country"]]]],
   modelData];

dataRFI = Dataset[dataWithRFI];

(* --- Sanity check: RFI must lie on [0, 1] -------------------------- *)
rfiValues = Cases[dataWithRFI,
   KeyValuePattern[{"RelFracIndex" -> r_?NumericQ}] :> r];
Print["RFI range: ", {Min[rfiValues], Max[rfiValues]},
  "  (must be within {0, 1})"];
Print["Countries with a usable RFI: ", Length[rfiValues]];
```

Scatterplot Generation

```
(* Points for migration vs. number of languages *)
migLangPoints =
  Cases[dataWithRFI,
   KeyValuePattern[{"MigrationRateNumber" -> mr_?NumericQ,
      "NumLanguages" -> nl_?NumericQ}] :> {mr, nl}];

plotMigLang = ListPlot[migLangPoints,
   FrameLabel -> {"Migration rate", "Number of languages"},
   PlotRange -> All, Frame -> True, ImageSize -> 500,
   LabelStyle -> {FontFamily -> "Times", FontSize -> 12}];
```

```
(* Migration vs religious fractionalization *)
ptsMigRFI =
  Cases[dataWithRFI,
   KeyValuePattern[{"MigrationRateNumber" -> mr_?NumericQ,
      "RelFracIndex" -> rfi_?NumericQ}] :> {mr, rfi}];

plotMigRFI = ListPlot[ptsMigRFI,
   FrameLabel -> {"Migration rate", "Religious fractionalization"},
   PlotRange -> All, Frame -> True, ImageSize -> 500,
   LabelStyle -> {FontFamily -> "Times", FontSize -> 12}];

(* --- Export at print resolution for the paper ----------------------- *)
(* Save this notebook inside the repo folder first, so NotebookDirectory *)
(* resolves. Otherwise replace figDir with an explicit path.           *)
figDir = FileNameJoin[{NotebookDirectory[], "figures"}];
If[! DirectoryQ[figDir], CreateDirectory[figDir]];

Export[FileNameJoin[{figDir, "mig-vs-languages.png"}], plotMigLang,
  ImageResolution -> 300];
Export[FileNameJoin[{figDir, "mig-vs-rfi.png"}], plotMigRFI,
  ImageResolution -> 300];

Print["Figures exported to: ", figDir];
```

Regression Models

```
(* ------------------------------------------------------------------- *)
(* IMPORTANT: LinearModelFit reads the RESPONSE from the LAST column of  *)
(* the data matrix. The preceding columns are matched, in order, to the  *)
(* variables listed in the third argument.                              *)
(*                                                                      *)
(* The original code emitted rows as {nl, mr, lgdp, ltv, lpop, la},      *)
(* which made LogArea the response and shifted every parameter label by  *)
(* one position. The response now comes last in each row.               *)
(* ------------------------------------------------------------------- *)

(* Data for language diversity model - response (nl) LAST *)
lingData =
  Cases[dataWithRFI,
   KeyValuePattern[{"NumLanguages" -> nl_?NumericQ,
      "MigrationRateNumber" -> mr_?NumericQ,
      "LogGDPPerCapita" -> lgdp_?NumericQ,
      "LogTradeValue" -> ltv_?NumericQ,
      "LogPopulation" -> lpop_?NumericQ,
      "LogArea" -> la_?NumericQ}]
    :> {mr, lgdp, ltv, lpop, la, nl}];

Print["N for the linguistic diversity model: ", Length[lingData]];

lmLang =
  LinearModelFit[
   lingData, {mr, lgdp, ltv, lpop, la}, {mr, lgdp, ltv, lpop, la}];

lmLang["ParameterTable"]
rSquared1 = lmLang["RSquared"];
Print["R Squared (languages): ", rSquared1];

(* Sanity check: fitted values should sit near the observed range *)
Print["Observed languages range: ",
  {Min[lingData[[All, -1]]], Max[lingData[[All, -1]]]}];
Print["Fitted languages range:  ",
  {Min[lmLang["PredictedResponse"]], Max[lmLang["PredictedResponse"]]}];


(* Data for religious fractionalization model - response (rfi) LAST *)
relFracData =
```

```
  Cases[dataWithRFI,
   KeyValuePattern[{"RelFracIndex" -> rfi_?NumericQ,
      "MigrationRateNumber" -> mr_?NumericQ,
      "LogGDPPerCapita" -> lgdp_?NumericQ,
      "LogTradeValue" -> ltv_?NumericQ,
      "LogPopulation" -> lpop_?NumericQ,
      "LogArea" -> la_?NumericQ}]
    :> {mr, lgdp, ltv, lpop, la, rfi}];

Print["N for the religious fractionalization model: ",
  Length[relFracData]];

lmRFI = LinearModelFit[
   relFracData, {mr, lgdp, ltv, lpop, la}, {mr, lgdp, ltv, lpop, la}];

lmRFI["ParameterTable"]
rSquared2 = lmRFI["RSquared"];
Print["R Squared (RFI): ", rSquared2];

(* Sanity check: fitted RFI should stay near [0, 1] *)
Print["Observed RFI range: ",
  {Min[relFracData[[All, -1]]], Max[relFracData[[All, -1]]]}];
Print["Fitted RFI range:   ",
  {Min[lmRFI["PredictedResponse"]], Max[lmRFI["PredictedResponse"]]}];
```

Collinearity Diagnostics

```
(* Log population, log area, and log trade value are all proxies for     *)
(* country size. Check how badly they overlap before interpreting any    *)
(* individual coefficient.                                               *)

predictorNames = {"mr", "lgdp", "ltv", "lpop", "la"};

Print["Predictor correlation matrix:"];
Print[TableForm[Correlation[lingData[[All, 1 ;; 5]]],
   TableHeadings -> {predictorNames, predictorNames}]];

(* Variance inflation factor: regress each predictor on the others.     *)
(* VIF > 5 is conventionally problematic, > 10 severe.                  *)
Clear[a, b, c, d]; vifVars = {a, b, c, d};
vifs = Table[
   Module[{cols, sub, m},
    cols = lingData[[All, 1 ;; 5]];
    sub = Transpose[Append[Drop[Transpose[cols], {j}], cols[[All, j]]]];
    m = LinearModelFit[sub, vifVars, vifVars];
    1/(1 - m["RSquared"])], {j, 5}];

Print["Variance inflation factors:"];
Print[TableForm[{predictorNames, vifs}]];
```

Reduced Specifications

```
(* With VIFs above 25, the full-model coefficients are not              *)
(* interpretable. Drop log population and log trade value, keeping      *)
(* migration rate, log GDP per capita, and log area.                    *)

(* Log area alone, against the raw count *)
areaOnly = LinearModelFit[lingData[[All, {5, 6}]], {la}, {la}];
areaOnly["ParameterTable"]
Print["R Squared (languages, area only): ", areaOnly["RSquared"]];

(* Reduced count model: mr, lgdp, la *)
lingReduced = lingData[[All, {1, 2, 5, 6}]];
lmLangRed = LinearModelFit[lingReduced, {mr, lgdp, la}, {mr, lgdp, la}];
```

```
lmLangRed["ParameterTable"]
Print["R Squared (languages, reduced): ", lmLangRed["RSquared"]];

(* Confirm the reduction actually resolved the collinearity *)
Clear[a, b]; vr = {a, b};
Print["Reduced VIFs (mr, lgdp, la): ", Table[
   Module[{cols = lingReduced[[All, 1 ;; 3]], sub},
    sub = Transpose[Append[Drop[Transpose[cols], {j}], cols[[All, j]]]];
    1/(1 - LinearModelFit[sub, vr, vr]["RSquared"])], {j, 3}]];

(* Same treatment for religious fractionalization *)
rfiAreaOnly = LinearModelFit[relFracData[[All, {5, 6}]], {la}, {la}];
rfiAreaOnly["ParameterTable"]
Print["R Squared (RFI, area only): ", rfiAreaOnly["RSquared"]];

rfiReduced = relFracData[[All, {1, 2, 5, 6}]];
lmRFIRed = LinearModelFit[rfiReduced, {mr, lgdp, la}, {mr, lgdp, la}];
lmRFIRed["ParameterTable"]
Print["R Squared (RFI, reduced): ", lmRFIRed["RSquared"]];
```

Count Models for Linguistic Diversity

```
(* The number of languages is a count from 1 to 839 with heavy right     *)
(* skew, and raw-count OLS produces negative fitted values. Two         *)
(* alternatives were considered.                                         *)

(* --- Poisson: rejected -------------------------------------------- *)
glmLang = GeneralizedLinearModelFit[
   lingData, {mr, lgdp, ltv, lpop, la}, {mr, lgdp, ltv, lpop, la},
   ExponentialFamily -> "Poisson"];

glmLang["ParameterTable"]
Print["Poisson model AIC: ", glmLang["AIC"]];
Print["Linear model AIC:  ", lmLang["AIC"]];

(* Poisson assumes variance equals the mean. A ratio far above 1 means   *)
(* the standard errors above are badly understated and the model must    *)
(* not be reported. Observed here: ~86.5.                                *)
Print["Overdispersion ratio: ",
  Total[(lingData[[All, -1]] - glmLang["PredictedResponse"])^2/
     glmLang["PredictedResponse"]]/(Length[lingData] - 6)];

(* --- Log-transformed response: adopted ------------------------------ *)
(* Handles the skew, keeps the response positive, and gives             *)
(* coefficients interpretable as elasticities. Every country reports at  *)
(* least one language, so there are no zeros to adjust for.              *)

logLingData = Append[#[[1 ;; 5]], Log[#[[6]]]] & /@ lingData;

lmLogLang = LinearModelFit[logLingData, {mr, lgdp, ltv, lpop, la},
   {mr, lgdp, ltv, lpop, la}];
lmLogLang["ParameterTable"]
Print["R Squared (log languages, full): ", lmLogLang["RSquared"]];

(* Reduced log model - this is the specification reported in the paper *)
logLingReduced = logLingData[[All, {1, 2, 5, 6}]];
lmLogRed = LinearModelFit[logLingReduced, {mr, lgdp, la}, {mr, lgdp, la}];
lmLogRed["ParameterTable"]
Print["R Squared (log languages, reduced): ", lmLogRed["RSquared"]];
```